

# GET TRANSACTION DATA

**Note:**

This statement is only valid for transactions applied to Adabas databases, or to DL/I databases in a batch-oriented BMP region (in IMS environments only).

**GET TRANSACTION [DATA] operand1...**

Operand	Possible Structure					Possible Formats					Referencing Permitted	Dynamic Definition					
Operand1	S					A	N	P	I	F	B	D	T			yes	yes

Related Statements: END TRANSACTION

---

## Function

The GET TRANSACTION DATA statement is used to read the data saved with a previous END TRANSACTION statement.

GET TRANSACTION DATA does not create a processing loop.

Note for DL/I:

The GET TRANSACTION DATA statement retrieves checkpoint data saved by an END TRANSACTION statement.

## System Variable \*ETID

The content of the Natural system variable \*ETID identifies the transaction data to be retrieved from the database.

## Field Specification operand1

The sequence, lengths and formats of the fields used in the GET TRANSACTION DATA statement must be identical to the sequence, lengths and formats of the fields specified with the corresponding END TRANSACTION statement.

Note for DL/I:

The first operand1 must be an 8-byte checkpoint ID.

## No Transaction Data Stored

If the GET TRANSACTION DATA statement is issued and no transaction data are found, all fields specified in the GET TRANSACTION DATA statement will be filled with blanks regardless of format definition. Make sure that arithmetic operations are not performed on "empty" transaction data, because this would result in an abnormal termination of the program.

## Example

```
/* EXAMPLE 'GTREX1S': GET TRANSACTION DATA (STRUCTURED MODE)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
2 PERSONNEL-ID
2 NAME
2 FIRST-NAME
2 MIDDLE-I
2 CITY
1 #PERS-NR (A8) INIT <' '>
END-DEFINE
*****
GET TRANSACTION DATA #PERS-NR
IF #PERS-NR NE ''
  WRITE 'LAST TRANSACTION PROCESSED FROM PREVIOUS SESSION' #PERS-NR
END-IF
*****
REPEAT
*****
  INPUT 10X 'ENTER PERSONNEL NUMBER TO BE UPDATED:' #PERS-NR
  IF #PERS-NR = ''
    STOP
  END-IF
*****
  FIND EMPLOY-VIEW WITH PERSONNEL-ID = #PERS-NR
  IF NO RECORDS FOUND
    REINPUT 'NO RECORD FOUND'
  END-NOREC
  INPUT (AD=M) PERSONNEL-ID (AD=O)
    / NAME
    / FIRST-NAME
    / CITY
  UPDATE
  END TRANSACTION #PERS-NR
END-FIND
*****
END-REPEAT
END
```